1. Sur $]0,1]$, $-f_1$ est continue et AVP et $-f_1(t) \underset{t\to 0}{\sim} -\ln t$ donc $f_1 \in L_1(]0,1])$.

   Sur $[1,+\infty[$, f_1 est continue et AVP et $f_1(t) \underset{t\to+\infty}{\sim} \dfrac{\ln t}{t^2}$ donc $f_1 \in L_1([1,+\infty[)$ (Int. de Bertrand).

2. $\displaystyle\int_0^{+\infty} f_1 = \ln 2 \int_0^{+\infty} \frac{\mathrm{d}u}{(2+u)(1+u)} - \int_0^{+\infty} f_1$ donc $\displaystyle\int_0^{+\infty} f_1 = \frac{\ln 2}{2}\int_0^{+\infty}\left(\frac{1}{1+u} - \frac{1}{2+u}\right)\mathrm{d}u$ donne bien $\dfrac{\ln^2 2}{2}$.

3. Pour f_3, par IPP $\displaystyle\int_0^{+\infty} f_3 = \int_0^{+\infty} \frac{\mathrm{d}t}{t(t+a)} = (1/a)\int_0^{+\infty}\left(\frac{1}{t} - \frac{1}{t+a}\right)\mathrm{d}t$ ce qui donne bien $\dfrac{\ln a}{a}$.

4. $\alpha = -\beta = \dfrac{1}{a_{n+1} - a_n}$ donc $I_{n+1}(a) = \alpha(I_n(a) - I_n(a'))$.

O17-C035

```
> restart;
> f:=n->ln(t)/product(t+k,k=1..n+1);
```

$$f := n \to \frac{\ln(t)}{\prod_{k=1}^{n+1}(t+k)}$$

```
> int1:=int(f(1),t=0..infinity);
```

$$int1 := \frac{1}{2}\ln(2)^2$$

```
> with(IntegrationTools):
  V := Int(f(1),t=0..infinity);
> V1:=Change(V, t=2/u);
```

$$V := \int_0^\infty \frac{\ln(t)}{(t+1)(t+2)}\,dt$$

$$V1 := -\int_0^\infty -\frac{\ln(2)-\ln(u)}{(2+u)(1+u)}\,du$$

```
> int(ln(2)/(1+u)/(2+u)/2,u=0..infinity);
```

$$\frac{1}{2}\ln(2)^2$$

```
> int2:=int(f(2),t=0..infinity);
```

$$int2 := -\frac{1}{4}\ln(3)^2 + \frac{1}{2}\ln(2)^2$$

```
> assume (a>0);int(ln(t)/(t+a)^2,t=0..infinity);
```

$$\frac{\ln(a\sim)}{a\sim}$$

```
> J:=n->int(f(n),t=0..infinity);seq(J(n),n=1..10);
```

$$J := n \to \int_0^\infty \mathrm{f}(n)\,dt$$

$$\frac{1}{2}\ln(2)^2,\ -\frac{1}{4}\ln(3)^2+\frac{1}{2}\ln(2)^2,\ -\frac{1}{4}\ln(3)^2+\frac{7}{12}\ln(2)^2,\ -\frac{1}{8}\ln(3)^2+\frac{5}{12}\ln(2)^2-\frac{1}{48}\ln(5)^2,$$
$$\frac{23}{120}\ln(2)^2+\frac{1}{120}\ln(2)\ln(3)-\frac{3}{80}\ln(3)^2-\frac{1}{48}\ln(5)^2,\ \frac{23}{360}\ln(2)^2+\frac{1}{120}\ln(2)\ln(3)-\frac{1}{160}\ln(3)^2-\frac{1}{1440}\ln(7)^2-\frac{1}{96}\ln(5)^2,$$
$$\frac{59}{3360}\ln(2)^2-\frac{1}{1440}\ln(7)^2+\frac{1}{240}\ln(2)\ln(3)-\frac{1}{288}\ln(5)^2,$$
$$\frac{1}{3360}\ln(3)^2+\frac{1}{224}\ln(2)^2+\frac{1}{720}\ln(2)\ln(3)-\frac{1}{1152}\ln(5)^2-\frac{1}{2880}\ln(7)^2,$$
$$\frac{199}{181440}\ln(2)^2+\frac{1}{362880}\ln(2)\ln(5)-\frac{25}{145152}\ln(5)^2-\frac{1}{8640}\ln(7)^2+\frac{1}{13440}\ln(3)^2+\frac{1}{2880}\ln(2)\ln(3),\ -\frac{1}{7257600}\ln(11)^2$$
$$+\frac{229}{907200}\ln(2)^2+\frac{1}{362880}\ln(2)\ln(5)-\frac{1}{36288}\ln(5)^2+\frac{1}{14400}\ln(2)\ln(3)+\frac{1}{268800}\ln(3)^2-\frac{1}{34560}\ln(7)^2$$

```
> g:=n->ln(t)/product(t+a[k],k=1..n+1);
```

$$g := n \to \frac{\ln(t)}{\prod_{k=1}^{n+1}(t+a_k)}$$

```
> assume(a[1]>0,a[2]>0);intgen1:=int(g(1),t=0..infinity);
```

$$intgen1 := \frac{1}{2}\,\frac{\ln(a\sim_1)^2-\ln(a\sim_2)^2}{-a\sim_2+a\sim_1}$$

```
> convert(1/(t+b)/(t+a),parfrac,t);
```

$$-\frac{1}{(a\sim-b)(t+a\sim)}+\frac{1}{(a\sim-b)(t+b)}$$

```
> In:=proc(a,n)
  local alpha,b,c;
  if n=1 then
            return int(ln(t)/(t+a[1])/(t+a[2]),t=0..infinity)
  else
            alpha:=1/(a[n+1]-a[n]);
            b:=[seq(a[k],k=1..n)];
            c:=[seq(a[k],k=1..n-1),a[n+1]];
            return((In(b,n-1)-In(c,n-1))*alpha)
  fi;
  end;
```

In := **proc**(*a*, *n*)
local α, *b*, *c*;
    if $n = 1$ **then return** int(ln(*t*) / ((*t* + *a*[1])*(*t* + *a*[2])), *t* = 0 .. ∞)
    else
        α := 1 / (*a*[*n* + 1] − *a*[*n*]);
        b := [seq(*a*[*k*], *k* = 1 .. *n*)];
        c := [seq(*a*[*k*], *k* = 1 .. *n* − 1), *a*[*n* + 1]];
        return (In(*b*, *n* − 1) − In(*c*, *n* − 1))*α
    end if
end proc

```
> In([a,b],1);
```

$$\begin{cases} \mathit{undefined} & \mathrm{And}(a\sim < 0,\ b < 0) \\ \mathit{undefined} & a\sim < 0 \\ \mathit{undefined} & b < 0 \\ \dfrac{1}{2}\,\dfrac{\ln\left(\dfrac{1}{a\sim}\right)^2 - \ln\left(\dfrac{1}{b}\right)^2}{a\sim - b} & \mathit{otherwise} \end{cases}$$

```
> In([1,2,3],2);
```

$$-\frac{1}{4}\ln(3)^2 + \frac{1}{2}\ln(2)^2$$

```
> seq(In([seq(k,k=1..n+1)],n),n=1..10);
```

$$\frac{1}{2}\ln(2)^2,\ -\frac{1}{4}\ln(3)^2 + \frac{1}{2}\ln(2)^2,\ -\frac{1}{4}\ln(3)^2 + \frac{7}{12}\ln(2)^2,\ -\frac{1}{8}\ln(3)^2 + \frac{5}{12}\ln(2)^2 - \frac{1}{48}\ln(5)^2,$$

$$\frac{23}{120}\ln(2)^2 + \frac{1}{120}\ln(2)\ln(3) - \frac{3}{80}\ln(3)^2 - \frac{1}{48}\ln(5)^2,\ \frac{23}{360}\ln(2)^2 + \frac{1}{120}\ln(2)\ln(3) - \frac{1}{160}\ln(3)^2 - \frac{1}{1440}\ln(7)^2 - \frac{1}{96}\ln(5)^2,$$

$$\frac{59}{3360}\ln(2)^2 - \frac{1}{1440}\ln(7)^2 + \frac{1}{240}\ln(2)\ln(3) - \frac{1}{288}\ln(5)^2,$$

$$\frac{1}{3360}\ln(3)^2 + \frac{1}{224}\ln(2)^2 + \frac{1}{720}\ln(2)\ln(3) - \frac{1}{1152}\ln(5)^2 - \frac{1}{2880}\ln(7)^2,$$

$$\frac{199}{181440}\ln(2)^2 + \frac{1}{362880}\ln(2)\ln(5) - \frac{25}{145152}\ln(5)^2 - \frac{1}{8640}\ln(7)^2 + \frac{1}{13440}\ln(3)^2 + \frac{1}{2880}\ln(2)\ln(3),\ -\frac{1}{7257600}\ln(11)^2$$

$$+\frac{229}{907200}\ln(2)^2 + \frac{1}{362880}\ln(2)\ln(5) - \frac{1}{36288}\ln(5)^2 + \frac{1}{14400}\ln(2)\ln(3) + \frac{1}{268800}\ln(3)^2 - \frac{1}{34560}\ln(7)^2$$

```
>
```